

16 Linear Differential Equations with Constant Coefficients

This chapter provides examples showing how Maple V can be used to solve higher order linear differential equations with constant coefficient. Some of the examples illustrate how Maple V can be used to solve problems the same way that one does when solving differential equations by hand. Other examples are solved using the Maple V command **dsolve**. Generally, when you are solving these kind of problems you will probably use **dsolve**. We have included the other procedures here to reenforce the methods that you learn in the traditional course and to illustrate how Maple V can simplify complex problems that involve lengthy computations.

16.1 Solving Homogeneous Equations

The first example will be to solve a fourth order linear differential equation with constant coefficients. The steps used here are the same as one might use when working a problem by hand. In this example we solve the following differential equation.

$$\frac{d^4}{dx^4} y(x) + 3 \frac{d^3}{dx^3} y(x) - 16 \frac{d^2}{dx^2} y(x) + 12 \frac{d}{dx} y(x) = 0$$

Subsequently, we wish to find the solution of the differential equation which satisfies the initial values:

$$y(0) = 2, y'(0) = 0, y''(0) = 4, y'''(0) = 0.$$

We enter the differential equation and assign it to the variable “deq1.”

```
> deq1 := diff(y(x), x$4) + 3*diff(y(x), x$3) - 16*diff(y(x), x$2) +
12*diff(y(x), x) = 0;
```

$$deq1 := \left(\frac{\partial^4}{\partial x^4} y(x) \right) + 3 \left(\frac{\partial^3}{\partial x^3} y(x) \right) - 16 \left(\frac{\partial^2}{\partial x^2} y(x) \right) + 12 \left(\frac{\partial}{\partial x} y(x) \right) = 0$$

The characteristic equation is obtained by replacing the n th order derivative by r^n .

```
> ceq1 := subs({diff(y(x), x$4)=r^4, diff(y(x), x$3)=r^3, diff(y(x), x$2)=r^2,
diff(y(x), x)=r}, deq1);
```

$$ceq1 := r^4 + 3r^3 - 16r^2 + 12r = 0$$

The first step in solving the problem “deq1” is to find the roots of the characteristic equation.

Note: The answer or result provided by Maple V in the next two steps is a sequence of Maple objects separated by commas. Such a construction is called an **expression sequence**.

```
> solve(ceq1, r);
```

$$0, 1, 2, -6$$

```
> roots1 := ";
```

$$roots1 := 0, 1, 2, -6$$

Once we know the characteristic roots we can obtain a basis for the general solution. When we enclose a sequence like 0, 1, 2, -6 with square brackets, [...], we create a Maple V object known as a list. A Maple V **list** preserves order of data in the way you specify it to be. Next we apply a Maple function known as **map** to the list [0, 1, 2, -6]. This Maple function applies the same operation to each component in a data structure. The **map** command is a kind of “advanced” Maple command and you may need to invoke

> ? map

in order to obtain help. We now apply **map** twice to obtain a base of solutions for the homogeneous solution.

> map((x,y) -> x*y, [croots1], x);

$$[0, x, 2x, -6x]$$

> base1 := map(exp, ");

$$base1 := [1, e^x, e^{(2x)}, e^{(-6x)}]$$

We can now write the general solution:

> ygeneral1 := sum(c[i]*base1[i], i=1..4);

$$ygeneral1 := c_1 + c_2 e^x + c_3 e^{(2x)} + c_4 e^{(-6x)}$$

In order to solve the initial value problem we need to solve for the constants by solving the equations:

$$y(0) = 2, \quad y'(0) = 0, \quad y''(0) = 4, \quad y'''(0) = 0.$$

Maple V commands that do this are as follows:

> y0 := subs(x=0, ygeneral1);

$$y0 := c_1 + c_2 e^0 + c_3 e^0 + c_4 e^0$$

> dy0 := subs(x=0, diff(ygeneral1, x));

$$dy0 := c_2 e^0 + 2c_3 e^0 - 6c_4 e^0$$

> ddy0 := subs(x=0, diff(ygeneral1, x\$2));

$$ddy0 := c_2 e^0 + 4c_3 e^0 + 36c_4 e^0$$

> dddy0 := subs(x=0, diff(ygeneral1, x\$3));

$$dddy0 := c_2 e^0 + 8c_3 e^0 - 216c_4 e^0$$

> pcont1 := solve({y0=2, dy0=0, ddy0=4, dddy0=0}, {c[1], c[2], c[3], c[4]});

$$pcont1 := \left\{ c_3 = \frac{5}{4}, c_2 = \frac{-16}{7}, c_1 = 3, c_4 = \frac{1}{28} \right\}$$

Using the above result we can now write the particular solution to the initial value problem.


> ylp := subs(pcont1, ygeneral1);

$$y1p := 3 - \frac{16}{7}e^x + \frac{5}{4}e^{(2x)} + \frac{1}{28}e^{(-6x)}$$

We will give an example to show you how to include the graph of the function, yc , within the worksheet. First execute the following command:

```
> plot(y1p, x=0..3);
```

Maple V should respond by opening a window with a graph of the desired function. It will look something like the following:



You may leave the window there while you execute other commands. Use the mouse to move the window to another section of the screen. Note that you may change the size of the window by placing the cursor on the lower right corner, holding down the left button on the mouse and moving the mouse. Try it. You can also copy the plot to your Maple Worksheet. That's how we got the graph above. All you do (on a UNIX machine with X-windows) is:

1. Select edit in the menu bar of the plotting window with the left button on the mouse.
2. With the edit item open select the copy option, again with the left mouse button. This temporarily puts a copy of the graph in the "clipboard."
3. Move the cursor to some point in the Maple Worksheet, click the left mouse button.
4. Select the edit item from the menu bar of the Maple Worksheet and select paste when the menu appears.

Try this now by attempting to put a copy of the plot you have in the window. If you are using a Windows base PC or a Macintosh then simply do the usual copy and pasting between windows.

Hopefully you were successful. You can use this technique to include graphs in assignments. So practice this until you can perform the procedure successfully. Get help if you're having trouble.

The next example involves a characteristic equation with complex roots. Consider the following equation.

$$\frac{d^4}{dx^4}y(x) - 5\frac{d^3}{dx^3}y(x) + 3\frac{d^2}{dx^2}y(x) + 19\frac{d}{dx}y(x) - 30y(x) = 0$$

Again we define the differential equation as the variable "deq2."

```
> deq2 := diff(y(x), x$4) - 5*diff(y(x), x$3) + 3*diff(y(x), x$2) + 19*diff(y(x), x)
- 30*y(x) = 0;
```

$$deq2 := \left(\frac{\partial^4}{\partial x^4} y(x) \right) - 5 \left(\frac{\partial^3}{\partial x^3} y(x) \right) + 3 \left(\frac{\partial^2}{\partial x^2} y(x) \right) + 19 \left(\frac{\partial}{\partial x} y(x) \right) - 30 y(x) = 0$$

Next, we create the characteristic equation.

```
> subs({diff(y(x), x$4)=r^4, diff(y(x), x$3)=r^3, diff(y(x), x$2)=r^2,
diff(y(x), x) = r, y(x)=1}, deq2);
```

$$r^4 - 5r^3 + 3r^2 + 19r - 30 = 0$$

```
> ceq2 := ";
```

$$ceq2 := r^4 - 5r^3 + 3r^2 + 19r - 30 = 0$$

Next we find the characteristic roots.

```
> roots2 := solve(ceq2, r);
```

$$roots2 := 3, -2, 2 + I, 2 - I$$

Since two of the roots are complex, the base will involve sine and cosine functions.

```
> base2 := [exp(3*x), exp(-2*x), exp(2*x)*cos(x), exp(2*x)*sin(x)];
```

$$base2 := [e^{(3x)}, e^{(-2x)}, e^{(2x)} \cos(x), e^{(2x)} \sin(x)]$$

The general solution is:

```
> ygeneral2 := sum(c[i]*base2[i], i=1..4);
```

$$ygeneral2 := c_1 e^{(3x)} + c_2 e^{(-2x)} + c_3 e^{(2x)} \cos(x) + c_4 e^{(2x)} \sin(x)$$

Now, if we desire a solution of deq that satisfies the initial conditions,

$$y(0) = -2, \quad y'(0) = 0, \quad y''(0) = 1, \quad y'''(0) = 14,$$

we proceed as follows.

```
> y0 := subs(x=0, ygeneral2);
```

$$y0 := c_1 e^0 + c_2 e^0 + c_3 e^0 \cos(0) + c_4 e^0 \sin(0)$$

```
> y0 := simplify(");
```

$$y0 := c_1 + c_2 + c_3$$

```
> dy0 := simplify(subs(x=0,diff(ygeneral2,x)));
```

$$dy0 := 3c_1 - 2c_2 + 2c_3 + c_4$$

```
> ddy0 := simplify(subs(x=0,diff(ygeneral2,x$2)));
```

$$ddy0 := 9c_1 + 4c_2 + 3c_3 + 4c_4$$

```
> dddy0 := simplify(subs(x=0,diff(ygeneral2,x$3)));
```

$$dddy0 := 27c_1 - 8c_2 + 2c_3 + 11c_4$$

```
> solve({y0=-2,dy0=0,ddy0=1,dddy0=14},{c[1],c[2],c[3],c[4]});
```

$$\left\{ c_2 = \frac{-37}{85}, c_3 = \frac{-13}{17}, c_4 = \frac{52}{17}, c_1 = \frac{-4}{5} \right\}$$

```
> subs(",ygeneral2);
```

$$-\frac{4}{5}e^{(3x)} - \frac{37}{85}e^{(-2x)} - \frac{13}{17}e^{(2x)}\cos(x) + \frac{52}{17}e^{(2x)}\sin(x)$$

We now use the Maple command **dsolve** to solve these same problems in one step.

The general solution for the first differential equation, “deq1”, can be obtained as follows:

```
> dsolve(deq1,y(x));
```

$$y(x) = _C1 + _C2 e^x + _C3 e^{(2x)} + _C4 e^{(-6x)}$$

It is just as easy to find the solution of the initial value problem:

```
> dsolve({deq1,y(0)=2,D(y)(0)=0,D(D(y))(0)=4,D(D(D(y)))(0)=0},y(x));
```

$$y(x) = 3 - \frac{16}{7}e^x + \frac{5}{4}e^{(2x)} + \frac{1}{28}e^{(-6x)}$$

We will now solve the second problem rewriting it so as to illustrate the use of the **D** notation.

```
> deq2 := (D@@4)(y)(x) - 5*(D@@3)(y)(x) + 3*(D@@2)(y)(x) + 19*D(y)(x)
-30*y(x) = 0;
```

$$deq2 := D^{(4)}(y)(x) - 5D^{(3)}(y)(x) + 3D^{(2)}(y)(x) + 19D(y)(x) - 30y(x) = 0$$

```
> dsolve({deq2,y(0)= -2,(D(y))(0)=0,(D@@2)(y)(0)=1,(D@@3)(y)(0)=14},y(x));
```

$$y(x) = -\frac{4}{5}e^{(3x)} - \frac{37}{85}e^{(-2x)} - \frac{13}{17}e^{(2x)}\cos(x) + \frac{52}{17}e^{(2x)}\sin(x)$$

16.2 Solving Non Homogeneous Linear Equations

Next we illustrate how Maple can be used to find a particular solution of a non homogeneous linear equation by the method of undetermined coefficients. We will find the general solution of the following equation:

$$\frac{d^3}{dx^3} y(x) - 4 \frac{d^2}{dx^2} y(x) + \frac{d}{dx} y(x) + 6y(x) = (x^3 - 4x + 2)e^x$$

First we enter the equation and then we proceed as if we were solving the equation by hand.

```
> deq3 := diff(y(x), x$3) - 4*diff(y(x), x$2) + diff(y(x), x) + 6*y(x) = (x^3 - 4*x + 2)*exp(x);
```

$$deq3 := \left(\frac{\partial^3}{\partial x^3} y(x) \right) - 4 \left(\frac{\partial^2}{\partial x^2} y(x) \right) + \left(\frac{\partial}{\partial x} y(x) \right) + 6y(x) = (x^3 - 4x + 2)e^x$$

The first step is to find the general solution of the associated homogeneous equation. Thus we obtain the characteristic equation.

```
> ceqn3 := subs({diff(y(x), x$3)=r^3, diff(y(x), x$2)=r^2, diff(y(x), x)=r,
y(x)=1}, lhs(deq3))=0;
```

$$ceqn3 := r^3 - 4r^2 + r + 6 = 0$$

```
> roots3 := [solve(ceqn3)];
```

$$roots3 := [2, 3, -1]$$

```
> base3 := map((s,t)->exp(s*t), roots3, x);
```

$$base3 := [e^{(2x)}, e^{(3x)}, e^{(-x)}]$$

Now the general solution of the homogeneous equation is

```
> yc3 := sum(c[i]*base3[i], i=1..3);
```

$$yc3 := c_1 e^{(2x)} + c_2 e^{(3x)} + c_3 e^{(-x)}$$

Next we find a particular solution by undetermined coefficients. In this example we define yp as a function of x, a, b, c, and d.

```
> yp3 := (x, a, b, c, d) -> (a*x^3 + b*x^2 + c*x + d)*exp(x);
```

$$yp3 := (x, a, b, c, d) \rightarrow (ax^3 + bx^2 + cx + d)e^x$$

This is substituted into the differential equation, deq3, and simplified and conditions on the constants a, b, c, and d are obtained.

```
> deq3a := subs(y(x)=yp3(x, a, b, c, d), deq3);
```

$$\begin{aligned} \text{deq3a} &:= \left(\frac{\partial^3}{\partial x^3} \%1 \right) - 4 \left(\frac{\partial^2}{\partial x^2} \%1 \right) + \left(\frac{\partial}{\partial x} \%1 \right) + 6 \%1 = (x^3 - 4x + 2) e^x \\ \%1 &:= (ax^3 + bx^2 + cx + d) e^x \end{aligned}$$

```
> eq := simplify(expand(deq3a/exp(x)));
```

$$\text{eq} := 6a - 6ax - 2b - 12ax^2 - 8bx - 4c + 4ax^3 + 4bx^2 + 4cx + 4d = x^3 - 4x + 2$$

```
> left := lhs(eq)-rhs(eq);
```

$$\text{left} := 6a - 6ax - 2b - 12ax^2 - 8bx - 4c + 4ax^3 + 4bx^2 + 4cx + 4d - x^3 + 4x - 2$$

We must collect all coefficients of the same power of x and equate to zero.

```
> left := collect(",x);
```

$$\text{left} := (4a - 1)x^3 + (-12a + 4b)x^2 + (4 - 8b - 6a + 4c)x + 6a - 4c - 2b - 2 + 4d$$

Finally the constants a,b,c, and d can now be solved.

```
> solve({coeff(left,x^3)=0,coeff(left,x^2)=0,coeff(left,x)=0,coeff(left,x,0)=0}, {a,b,c,d});
```

$$\left\{ a = \frac{1}{4}, b = \frac{3}{4}, c = \frac{7}{8}, d = \frac{11}{8} \right\}$$

The particular solution is given by

```
> ypart3 := subs(",yp3(x,a,b,c,d));
```

$$\text{ypart3} := \left(\frac{1}{4}x^3 + \frac{3}{4}x^2 + \frac{7}{8}x + \frac{11}{8} \right) e^x$$

The above function is a particular solution obtained by undetermined coefficients. The following steps represent a check.

```
> subs(y(x)=ypart3,deq3);
```

$$\begin{aligned} \left(\frac{\partial^3}{\partial x^3} \%1 \right) - 4 \left(\frac{\partial^2}{\partial x^2} \%1 \right) + \left(\frac{\partial}{\partial x} \%1 \right) + 6 \%1 &= (x^3 - 4x + 2) e^x \\ \%1 &:= \left(\frac{1}{4}x^3 + \frac{3}{4}x^2 + \frac{7}{8}x + \frac{11}{8} \right) e^x \end{aligned}$$

```
> simplify(");
```

$$2e^x - 4e^x x + e^x x^3 = (x^3 - 4x + 2)e^x$$

```
> collect(",exp(x));
```

$$(x^3 - 4x + 2)e^x = (x^3 - 4x + 2)e^x$$

Thus the solution, `ypart3`, is indeed a solution.

The next example is an illustration of variation of parameters. In your class you probably use this method only for second order equations. Maple allows you to work harder problems.

```
> deq4 := diff(y(x),x$3)-4*diff(y(x),x$2)+diff(y(x),x)+6*y(x) = cos(x)^2;
```

$$deq4 := \left(\frac{\partial^3}{\partial x^3} y(x) \right) - 4 \left(\frac{\partial^2}{\partial x^2} y(x) \right) + \left(\frac{\partial}{\partial x} y(x) \right) + 6y(x) = \cos(x)^2$$

As was seen above, a general solution for the homogeneous system associated with the equation is

$$y_h := c_1 e^{2x} + c_2 e^{-x} + c_3 e^{3x}.$$

Thus we seek functions $u_1(x)$, $u_2(x)$, $u_3(x)$ such that

$$y_p = u_1(x) e^{2x} + u_2(x) e^{-x} + u_3(x) e^{3x}$$

is a particular solution. The system to solve is

$$e^{2x} u_1' + e^{-x} u_2' + e^{3x} u_3' = 0,$$

$$2e^{2x} u_1' - e^{-x} u_2' + 3e^{3x} u_3' = 0,$$

$$4e^{2x} u_1' + e^{-x} u_2' + 9e^{3x} u_3' = \cos(x)^2.$$

Let us derive this method using Maple for this situation. We have already found a base of solutions for the homogeneous equations, we reintroduce them now in the variable base.

```
> base := [exp(2*x), exp(-x), exp(3*x)];
```

$$base := [e^{(2x)}, e^{(-x)}, e^{(3x)}]$$

Now we set up the equations in the same way that we would do it if we were solving the equation by hand.

```
> U := [u[1], u[2], u[3]];
```

$$U := [u_1, u_2, u_3]$$

```
> i := 'i': yp := sum(U[i](x)*base[i], i=1..3);
```


$$yp := u_1(x)e^{(2x)} + u_2(x)e^{(-x)} + u_3(x)e^{(3x)}$$

The problem is to solve for the functions u_1 , u_2 , and u_3 so that yp satisfies $deq4$. We first must solve three equations with three unknowns, the derivatives of the functions u_1 , u_2 , u_3 .

```
> i := 'i': eq1 := sum(diff(U[i](x), x) * base[i], i=1..3) = 0;
```

$$eq1 := \left(\frac{\partial}{\partial x} u_1(x) \right) e^{(2x)} + \left(\frac{\partial}{\partial x} u_2(x) \right) e^{(-x)} + \left(\frac{\partial}{\partial x} u_3(x) \right) e^{(3x)} = 0$$

```
> i := 'i': eq2 := sum(diff(U[i](x), x) * simplify(diff(base, x)[i]), i=1..3) = 0;
```

$$eq2 := 2 \left(\frac{\partial}{\partial x} u_1(x) \right) e^{(2x)} - \left(\frac{\partial}{\partial x} u_2(x) \right) e^{(-x)} + 3 \left(\frac{\partial}{\partial x} u_3(x) \right) e^{(3x)} = 0$$

```
> i := 'i': eq3 := sum(diff(U(x), x)[i] * simplify(diff(base, x$2)[i]), i=1..3) = cos(x)^2;
```

$$eq3 := 4 \left(\frac{\partial}{\partial x} u_1(x) \right) e^{(2x)} + \left(\frac{\partial}{\partial x} u_2(x) \right) e^{(-x)} + 9 \left(\frac{\partial}{\partial x} u_3(x) \right) e^{(3x)} = \cos(x)^2$$

```
> i := 'i': dsol := solve({eq1, eq2, eq3}, {seq(diff(U(x), x)[i], i=1..3)});
```

$$dsol := \left\{ \frac{\partial}{\partial x} u_1(x) = -\frac{1}{3} \frac{\cos(x)^2}{e^{(2x)}}, \frac{\partial}{\partial x} u_2(x) = \frac{1}{12} \frac{\cos(x)^2}{e^{(-x)}}, \frac{\partial}{\partial x} u_3(x) = \frac{1}{4} \frac{\cos(x)^2}{e^{(3x)}} \right\}$$

Thus we have found the derivatives of the unknown functions and now we integrate to find the functions.

```
> i := 'i': sol := {seq(int(lhs(dsol[i]), x) = int(rhs(dsol[i]), x), i=1..3)};
```

$$\begin{aligned} sol := & \left\{ \begin{aligned} u_2(x) &= \frac{1}{60} (\cos(x) + 2 \sin(x)) e^x \cos(x) + \frac{1}{30} e^x, \\ u_3(x) &= \frac{1}{52} (-3 \cos(x) + 2 \sin(x)) e^{(-3x)} \cos(x) - \frac{1}{78} \frac{1}{(e^x)^3}, \\ u_1(x) &= -\frac{1}{24} (-2 \cos(x) + 2 \sin(x)) e^{(-2x)} \cos(x) + \frac{1}{24} \frac{1}{(e^x)^2} \end{aligned} \right\} \end{aligned}$$

Now we substitute back into yp to obtain the solution.

```
> yp := simplify(subs(sol, yp));
```

$$yp := \frac{11}{260} \cos(x)^2 - \frac{3}{260} \cos(x) \sin(x) + \frac{97}{1560}$$

Once a formula is obtained for a solution it is easy to find numerical values for it and its derivatives for arbitrary values of x .

```
> eval(subs(x=0, yp));
```

$$\frac{163}{1560}$$

```
> eval(subs(x=0,diff(yp,x)));
```

$$\frac{-3}{260}$$

```
> eval(subs(x=0,diff(yp,x$2)));
```

$$\frac{-11}{130}$$

The command **dsolve** will yield the same solution if we use the same initial conditions.

```
> dsolve({deq4,y(0)=163/1560,D(y)(0)=-3/260,(D@@2)(y)(0)=-11/130},y(x));
```

$$y(x) = \frac{11}{260} \cos(x)^2 - \frac{3}{260} \cos(x) \sin(x) + \frac{97}{1560}$$

We can also substitute yp into deq4.

```
> simplify(subs(y(x)=yp,deq4));
```

$$\cos(x)^2 = \cos(x)^2$$

Exercises 16.2 The purpose of this exercise is to obtain the general solution of the following differential equation

$$\frac{d^5}{dx^5}y(x) - 8 \frac{d^4}{dx^4}y(x) + 25 \frac{d^3}{dx^3}y(x) - 50 \frac{d^2}{dx^2}y(x) + 84 \frac{d}{dx}y(x) - 72 y(x) = x^2 e^{5x}$$

For convenience let $L(y)(x)$ be defined by

$$L(y)(x) = \frac{d^5}{dx^5}y(x) - 8 \frac{d^4}{dx^4}y(x) + 25 \frac{d^3}{dx^3}y(x) - 50 \frac{d^2}{dx^2}y(x) + 84 \frac{d}{dx}y(x) - 72 y(x)$$

1. Write the Maple V command that assigns the equation

$$L(y)(x) = 0$$

to the variable “deq”.

2. Write the Maple V commands that will allow you to write the characteristic polynomial for

$$L(y)(x) = 0.$$

3. Use Maple V to find the characteristic roots for

$$L(y)(x) = 0.$$

4. Write the general solution y_h for

$$L(y)(x) = 0.$$

5. Guess the form of a particular solution y_p of

$$L(y)(x) = x^2 \exp(5x)$$

to be used the in the method of undetermined coefficients.

6. Use Maple V to evaluate and to simplify

$$L(y_p)(x)$$

with these undetermined coefficients.

7. Use Maple V to find y_p (i.e. determine the coefficients).

8. Write the general solution to

$$L(y)(x) = x^2 \exp(5x).$$

9. Use the Maple V command **dsolve** to find the general solution to

$$L(y)(x) = x^2 \exp(5x).$$

10. Use the Maple command **dsolve** to find the solution of the initial value problem

$$L(y)(x) = x^2 \exp(5x), y(0) = 0, y'(0) = 0, y''(0) = 0, y'''(0) = 0, y^{(iv)}(0) = 0.$$

Then plot the graph of this solution, in the interval $[0,3]$, and paste it into your Worksheet.